Parametricity via Cohesion

C.B. Aberlé

Abstract: Parametricity is a key metatheoretic property of type systems, from which one is able to derive strong uniformity/modularity properties of proofs and programs within these systems. In recent years, various systems of dependent type theory have emerged with the aim of internalizing such parametric reasoning into their internal logic, employing various syntactic and semantic methods and concepts toward this end. This paper presents a first step toward the unification, simplification, and extension of these various methods for internalizing parametricity. Specifically, I argue that there is an essentially modal aspect of parametricity, which is intimately connected with the category-theoretic concept of cohesion.

On this basis, I describe a general categorical semantics for modal parametricity, develop a corresponding framework of axioms (with computational interpretations) in dependent type theory that can be used to internally represent and reason about such parametricity, and show this in practice by implementing these axioms in Agda and using them to verify parametricity properties of Agda programs. In closing, I sketch the outlines of a more general synthetic theory of parametricity, and consider potential applications in domains ranging from homotopy type theory to the analysis of linear programs.

The past, present & future of parametricity in type theory

Reynolds (1983) began his seminal introduction of the concept of *parametricity* with the declaration that "type structure is a syntactic discipline for enforcing levels of abstraction." In the ensuing decades, this idea of Reynolds' has been overwhelmingly vindicated by the success of type systems in achieving modularity and abstraction in domains ranging from programming to interactive theorem proving. Yet the fate of Reynolds' particular strategy for formalizing this idea in terms of logical relations is rather more ambiguous.

Reynolds' original analysis of parametricity targeted the polymorphic λ -calculus (aka System F). Intuitively, polymorphic programs in System F cannot inspect the types over which they are defined and so must behave *essentially the same* for all types at which they are instantiated. To make this intuitive idea precise, Reynolds posed an ingenious solution in terms of logical relations, whereby every System F type is equipped with a suitable binary relation on its inhabitants, such that all terms constructible in System F must preserve the relations defined on their component types. On this basis, Reynolds was able to establish many significant properties of the abstraction afforded by System F's type structure, e.g. all closed terms of type $\forall X.X \rightarrow X$ are extensionally equivalent to the identity function.

Reynolds' results have been extended to many type systems, programming languages, etc. However, as the complexity and expressivity of these systems increases, so too does the difficulty of defining parametricity relations for their types, and proving that the inhabitants of these types respect their corresponding relations. Moreover, the abstract, mathematical patterns underlying the definitions of such relations have not always been clear. The problem – from a certain point of view, at any rate – is that the usual theory of parametricity relations for type systems is "analytic," " i.e. defined in terms of the particular constructs afforded by those systems, when what would be more desirable is a "synthetic" theory that boils the requirements for parametricity down to a few axioms, such that any type theory satisfying these axioms would necessarily enjoy parametricity and the expected consequences thereof.

For this purpose, a fruitful strategy is to consider ways of *internalizing* parametricity into the internal logic of a dependent type theory. Parametricity proofs are typically conducted in the metatheory of a particular type system, but if that type system is expressive enough to encode mathematical propositions in its types – as is the case for dependent type theory – there exists the possibility of instead defining and proving parametricity properties within the system itself. By considering the sorts of constructs a dependent type theory must contain in order to afford such internal parametric reasoning, we may arrive at an axiomatic specification of parametricity itself, as desired.

However, such internalization of parametricity into dependent type theory poses challenges of its own, owing to the fact that internal parametricity may be applied to itself, generating a complex, higher-dimensional structure of relations on types (c.f. Bernardy & Moulin). Various type theories for internal parametricity have thus been proposed, each posing its own solution to taming this complexity. So far, however, the majority of these type theories have targeted only specific models (usually some appropriately-chosen presheaf categories) rather than an axiomatically-defined *class* of models, and in this sense the analysis of parametricity offered by these type theories remains *analytic*, rather than *synthetic*. This in particular limits any insight into whether and how these appraoches to internal parametricity may be related to one another, generalized, or simplified.

As a first step toward the unification, simplification, and generalization of these approaches to internal parametricity, I propose an analysis of parametricity in terms of the category-theoretic concept of *cohesion*. Cohesion here refers to a particular relation between categories whereby one is equipped with an adjoint string of modalities that together make its objects behave like abstract spaces, whose points are bound together by some sort of *cohesion* that serves to constrain the structure-preserving maps that exist between these spaces. Such a notion of cohesion is suspiciously reminiscent of the informal idea behind Reynolds' formulation of parametricity outlined above, particularly if one interprets the *cohesion* that binds types together as the structure of relations that inhere between them. Moreover, by inspection of the categorical models of extant type theories for internal parametricity, along with classical models of parametricity, one finds that many if not all of these exhibit cohesion, in this sense. The main contribution of this paper is thus to show that this basic setup of cohesion is essentially *all* that is needed to recover classical parametricity results internally in dependent type theory. As both an illustration and verification of this idea, this paper is also a literate Agda document wherein the axioms for and theorems resulting from such parametricity in terms of cohesion have been fully formalized and checked for validity.

```
{-# OPTIONS --rewriting --cohesion --flat-split --without-K #-} module parametricity-via-cohesion where
```

```
open import Agda.Primitive
open import Data.Empty
open import Agda.Builtin.Unit
open import Agda.Builtin.Sigma
open import Agda.Builtin.Equality
open import Agda.Builtin.Equality.Rewrite
```

Cohesion and Parametricity

The notion of *cohesion* as an abstract characterization of when one category (specifically a topos) behaves like a category of spaces defined over the objects of another, is due primarily to Lawvere. The central concept of axiomatic cohesion is an arrangement of four adjoint functors as in the following diagram:

$$\mathcal{E}$$

$$\Pi \vdash \Delta \vdash \Gamma \vdash \nabla$$

$$\mathcal{S}$$

where \mathcal{E} , \mathcal{S} are both topoi, Δ , ∇ are both fully faithful, and Π preserves finite products. Given such an arrangement, we think of the objects of \mathcal{E} as *spaces* and those of \mathcal{S} as *sets* (even if \mathcal{S} is not the category of sets), where Γ is the functor that sends a space to its set of points, Δ sends a set to the corresponding *discrete space*, ∇ sends a set to the corresponding *codiscrete space*, and Π sends a space to its set of connected components. These in turn induce a string of adjoint modalities on \mathcal{E} :

$$\int -|b-| \sharp$$

where $\int = \Delta \circ \Pi$ and $\sharp = \nabla \circ \Gamma$ are idempotent monads, $\flat = \Delta \circ \Gamma$ is an idempotent comonad, and both \flat and \sharp preserve finite limits.

A concrete example of cohesion comes from the category of reflexive graphs \mathbf{RGph} , which is cohesive over the category of sets \mathbf{Set} . Here, Γ is the functor that sends a reflexive graph to its set of vertices, Δ sends a set V to the "discrete" reflexive graph on V whose only edges are self-loops, ∇ sends V to the "codiscrete" graph where there is a unique edge between any pair of vertices, and Π sends a reflexive graph to its set of (weakly) connected components. It is worth noting, at this point, that many classical models of parametricity are based upon semantic interpretations of types as reflexive graphs, and this, I wish to argue is no accident, and the key property of reflexive graphs underlying such interpretations is precisely their cohesive structure. More generally, for any base topos S, we may construct its corresponding topos $\mathbf{RGph}(S)$ of internal reflexive graphs, which will similarly be cohesive over S, so we can in fact use the language of such internal reflexive graphs to derive parametricity results for any topos (or indeed, any ∞ -topos, as described below).

In fact, this same setup of cohesion is interpretable, *mutatis mutandis*, in the case where \mathcal{E} , \mathcal{S} are not topoi, but rather ∞ -*topoi*, i.e. models of homotopy type theory. This allows us to use the language of

¹Similarly, the category of bicubical sets is cohesive over that of cubical sets, a fact exploited by Nuyts, Devriese & Vezzossi in the semantics for their type theory for parametric quantifiers.

homotopy type theory – suitably extended with constructs for the above-described modalities (the b modality in particular, which, for technical reasons, cannot be axiomatized directly in ordinary HoTT) – to work *synthetically* with the structure of such a cohesive ∞ -topos. For present purposes, we accomplish this by working in Agda with the –-cohesion and –-flat-split flags enabled, along with –-without-K, which ensures compatibility with the treatment of propositional equality in HoTT.

I therefore begin by recalling some standard definitions from HoTT, which shall be essential in defining much of the structure to follow. Essentially all of these definitions have to do with the *identity type* former $_\equiv$ and its associated constructor refl : $\forall \{\ell\} \{A : Set \ell\} \{a : A\} \rightarrow a \equiv a$, as defined in the module Agda. Builtin. Equality:

module hott where

First of all, we have the operation of *transport*, which embodies the principle of *indiscernability of identicals* (i.e. if a0 = a1 and B a1 then B a0):

```
transp : \forall {\ell \kappa} {A : Set \ell} {a0 a1 : A} \rightarrow (B : A \rightarrow Set \kappa) \rightarrow (a0 \equiv a1) \rightarrow B a1 \rightarrow B a0 transp B refl b = b
```

The notion of *contractibility* then expresses the idea that a type is essentially uniquely inhabited.

```
isContr : \forall \{\ell\} (A : Set \ell) \rightarrow Set \ell isContr A = \Sigma A (\lambda a \rightarrow (b : A) \rightarrow a \equiv b)
```

Similarly, the notion of *equivalence* expresses the idea that a function between types has an *essentially unique* inverse. (Those familiar with HoTT may note that I use the *contractible fibres* definition of equivalence, as this shall be the most convenient to work with, for present purposes).

```
isEquiv : \forall \{\ell \ \kappa\} \{A : Set \ \ell\} \{B : Set \ \kappa\}
\rightarrow (A \rightarrow B) \rightarrow Set \ (\ell \sqcup \kappa)
isEquiv \{A = A\} \{B = B\} f =
(b : B) \rightarrow isContr \ (\Sigma \ A \ (\lambda \ a \rightarrow f \ a \equiv b))

mkInv : \forall \{\ell \ \kappa\} \{A : Set \ \ell\} \{B : Set \ \kappa\}
\rightarrow (f : A \rightarrow B) \rightarrow isEquiv \ f \rightarrow B \rightarrow A

mkInv f \in b = fst \ (fst \ (e b))
```

open hott

The reader familiar with HoTT may note that, so far, we have not included anything relating to the *univalence axiom*, arguably the characteristic axiom of HoTT. In fact, this is by design, as a goal of the current formalization is to assume only axioms that can be given straightforward computational interpretations that preserve the property that every term of the ambient type theory evaluates to a *canonical normal form* (canonicity), so that these axioms give a *constructive* and *computationally sound* interpretation of parametricity. While the univalence axiom *can* be given a computational interpretation compatible with canonicity, as in Cubical Type Theory, doing so is decidedly *not* a straightforward matter. Moreover, it turns out that the univalence axiom is largely unneeded in what follows, save

for demonstrating admissibility of some additional axioms which admit much more striaghtforward computational interpretations that are (conjecturally) compatible with canonicity. I thus shall have need for univalence only as a metatheoretic assumption. Nonetheless, for expositional purposes, it is useful to define univalence formally, for which purpose I include the following submodule:

module univalence where

To state univalence, we first define the operation that takes an identity to an equivalence between types:

The univalence axiom asserts that this map is itself an equivalence:

```
postulate axiom : \forall \{\ell\} \{A \ B : \mathbf{Set} \ \ell\} \rightarrow \mathbf{isEquiv} \ (\mathbf{idToEquiv} \ A \ B)
```

Intuitively, univalence allows us to convert equivalences between types into *identifications* of those types, which may then be transported over accordingly.

Having defined some essential structures of the language of HoTT, we may now proceed to similarly define some essential structures of the language of HoTT with cohesive modalities.

```
module cohesion where
```

Of principal importance here is the \flat modality, which (intuitively) takes a type A to its corresponding discretization $\flat A$, wherein all cohesion between points has been eliminated. However, in order for this operation to be well-behaved, A must not depend upon any variables whose types are not themselves discrete, in this sense. To solve this problem, the --cohesion flag introduces a new form of variable binding $\textcircled{b} \times (X)$, which metatheoretically asserts that X is an element of the discretization of X. In this case, we say that X is a *crisp* element of X.

With this notion in hand, we can define the \flat modality as an operation on *crisp* types:

```
data \flat {@\flat \ell : Level} (@\flat A : Set \ell) : Set \ell where con : (@\flat x : A) \rightarrow \flat A
```

As expected, the \flat modality is a comonad with the following counit operation ϵ :

```
\epsilon : {@b 1 : Level} {@b A : Set 1} \rightarrow b A \rightarrow A \epsilon (con x) = x
```

A crisp type is then *discrete* precisely when this map is an equivalence:

```
isDiscrete : \forall {@b \ell : Level} \rightarrow (@b A : Set \ell) \rightarrow Set \ell isDiscrete {\ell = \ell} A = isEquiv (\epsilon {\ell} {A})
```

open cohesion

Perhaps surprisingly, with these few definitions alone, we are already nearly in a position to derive parametricity theorems in Agda (and more generally, in any type theory supporting the above constructions). What more is needed is merely a way of detecting when the elements of a given type are *related*, or somehow bound together, by the cohesive structure of that type.

For this purpose, it is useful to take a geometric perspective upon cohesion, and correspondingly, parametricity. What we are after is essentially the *shape* of an abstract relation between points, and an object I in our cohesive topos \mathcal{E} (correspondingly, a type in our type theory) which *classifies* this shape in other objects (types) in that maps $I \to A$ correspond to such abstract relations between points in A. For this purpose, I propose that the *shape* of an abstract relation is nothing other than a *line segment*, i.e. two distinct points which are somehow *connected*. By way of concrete example, in the topos of reflexive graphs \mathbf{RGph} , the role of a classifier for this shape is played by the "walking edge" graph $\bullet \to \bullet$, consisting of two points and a single non-identity (directed) edge. More generally, using the language of cohesion, we can capture this notion of an abstract line segment in the following axiomatic characterization of I:

I is an object of \mathcal{E} that is strictly bipointed and weakly connected.

Unpacking the terms used in this characterization, we have the following:

- *Strictly bipointed* means that *I* is equipped with a choice of two elements $i_0, i_1 : I$, such that the proposition $(i_0 = i_1) \rightarrow \bot$ (i.e. $i_0 \neq i_1$) holds in the internal language of \mathcal{E} .
- Weakly connected means that the unit map $\eta: I \to \int I$ is essentially constant, in that it factors through a contractible object/type. Intuitively, this says that the image of I in $\int I$ essentially consists of a single connected component.

Note that the above-given example of the walking edge graph straightforwardly satisfies both of these requirements, as it consists of two distinct vertices belonging to a single (weakly) connected component. I also note in passing that, If the assumption of weak connectedness is strengthened to strong connectedness – i.e. the object/type $\int I$ is itself contractible – then the existence of such an object I as above is equivalent to Lawvere's axiom of sufficient cohesion, as proved by him.

We can begin to formalize this idea in Agda by postulating a type I with two given elements i_0 , i_1 :

```
postulate
I : Set<sub>0</sub>
i0 i1 : I
```

We could also, in principle, directly postulate strict bipointedness of I, as in the following module:

```
module strictBpt where postulate  \text{axiom} : \text{i0} \equiv \text{i1} \rightarrow \bot
```

However, this is in fact unnecessary, as this axiom will instead follow from an equivalent formulation introduced in the following subsection – hence why this axiom is postulated in a separate module.

On the other hand, we do not yet have the capability to postulate the axiom of connectedness as written above, since we have not yet formalized the \int modality. We could do so, but again, it is in fact

better for present purposes to rephrase this axiom in an equivalent form involving only the b modality, which can be done as follows:

A crisp type A is connected if and only if, for every *discrete* type B, any function $A \rightarrow B$ is essentially constant, in the sense of factoring through a contractible type.

To see that this equivalence holds: in one direction, assume that A is weakly connected. Then for any map $f: A \to B$, by the adjunction $\int \neg b$ and discreteness of B, there exist maps $f_b: A \to bB$ and $f^f: \int A \to B$, such that following commuting diagram commutes:

$$\begin{array}{cccc} A & -\eta \to \int A \\ \downarrow & & \downarrow \\ f_{\flat} & f & f^{\flat} \\ \psi & & \downarrow & \psi \\ \flat B & -\epsilon \to B \end{array}$$

Then since by assumption η factors through a contractible type, so does f.

In the other direction, assume that every map $f: A \to B$ is essentially constant, for every discrete type B. Then in particular, the map $\eta: A \to \int A$ is essentially constant, since $\int A$ is discrete (as it lies in the image of the *discretization* functor Δ).

Hence the property of I being weakly connected is just as much a particular relation between I and all discrete types as it is a property of I itself. Specifically, if we think of maps $I \to A$ as abstract relations or edges between elements of A, then weak connectedness of I equivalently says that all edges between elements of discrete types are constant.

In order to conveniently express this property in Agda, it shall therefore be prudent first to introduce some additional constructs for ergonomically handling edges, analogous to the definition of *path* types in Cubical type theory. For this purpose, I now introduce a corresponding notion of *edge types*.

Edge Types

In principle, given a, b: A, we could define the type of edges from a to b in A as the type $\Sigma f: I \to A.(f i_0 = a) \times (f i_1 = b)$. However, experience with such a naïve formalization shows that it incurs a high number of laborious transportations along equalities that should be easy enough to infer automatically. Hence I instead follow the approach taken by cubical type theory and related systems, such as simplicial type theory, and give an explicit axiomatization for *edge types*, with corresponding rewrite rules to apply the associated equalities automatically:

postulate

```
Edge : \forall \{\ell\} (A : I \rightarrow Set \ell) (a0 : A i0) (a1 : A i1) \rightarrow Set \ell
```

The introduction rule for edge types corresponds to function abstraction

```
eabs : \forall {\ell} {A : I \rightarrow Set \ell} \rightarrow (f : (i : I) \rightarrow A i) \rightarrow Edge A (f i0) (f i1)
```

and likewise, the elimination rule corresponds to function application.

```
eapp : \forall {\ell} {A : I \rightarrow Set \ell} {a0 : A i0} {a1 : A i1} \rightarrow (e : Edge A a0 a1) \rightarrow (i : I) \rightarrow A i
```

We may then postulate the usual β -law as an identity for this type, along with special identities for application to i0 and i1. All of these are made into rewrite rules, allowing Agda to apply them automatically, and thus obviating the need for excessive use of transport:

```
ebeta : \forall \ \{\ell\} \ \{A : I \to Set \ \ell\} \ (f : (i : I) \to A i)
\to (i : I) \to eapp \ (eabs \ f) \ i \equiv f \ i
{-# REWRITE ebeta #-}
eapp0 : \forall \ \{\ell\} \ \{A : I \to Set \ \ell\} \ \{a0 : A \ i0\} \ \{a1 : A \ i1\}
\to (e : Edge \ A \ a0 \ a1) \to eapp \ e \ i0 \equiv a0
{-# REWRITE eapp0 #-}
eapp1 : \forall \ \{\ell\} \ \{A : I \to Set \ \ell\} \ \{a0 : A \ i0\} \ \{a1 : A \ i1\}
\to (e : Edge \ A \ a0 \ a1) \to eapp \ e \ i1 \equiv a1
{-# REWRITE eapp1 #-}
```

(We could additionally postulate an η -law for edge types, analogous to the usual η -law for function types; however, this is unnecessary for what follows, and so I omit this assumption.)

With this formalization of edge types in hand, we can straightforwardly formalize the equivalent formulation of weak connectedness of I given above. For this purpose, we first define the map idToEdge that takes an identification $a \equiv b$ to an edge from a to b:

```
idToEdge : \forall {\ell} {A : Set \ell} {a b : A} \rightarrow a \equiv b \rightarrow Edge (\lambda _ \rightarrow A) a b idToEdge {a = a} refl = eabs (\lambda _ \rightarrow a)
```

A type A is *edge-discrete* if for all a, b : A the mape idToEdge is an equivalence:

```
isEdgeDiscrete : \forall {\ell} (A : Set \ell) \rightarrow Set \ell isEdgeDiscrete {\ell = \ell} A = {a b : A} \rightarrow isEquiv (idToEdge {\ell} {A} {a} {b})
```

We then postulate the following axioms:

postulate

```
edgeConst1 : \forall {@b \ell : Level} {@b A : Set \ell} {a b : A} \rightarrow isDiscrete A \rightarrow (e : Edge (\lambda \_ \rightarrow A) a b) \rightarrow \Sigma (a \equiv b) (\lambda p \rightarrow idToEdge p \equiv e) edgeConst2 : \forall {@b \ell : Level} {@b A : Set \ell} {a b : A} \rightarrow (dA : isDiscrete A) \rightarrow (e : Edge (\lambda \_ \rightarrow A) a b) \rightarrow (q : a \equiv b) \rightarrow (r : idToEdge q \equiv e) \rightarrow edgeConst1 dA e \equiv (q , r)
```

which together imply that, if *A* is discrete, then it is *edge-discrete*:

```
isDisc→isEDisc : \forall {@b \ell : Level} {@b A : Set \ell} \rightarrow isDiscrete A \rightarrow isEdgeDiscrete A isDisc→isEDisc dA e = (edgeConst1 dA e , \lambda (p , q) \rightarrow edgeConst2 dA e p q)
```

As it stands, we have not yet given a procedure for evaluating the axioms edgeConst1 and

edgeConst2 when they are applied to canonical forms, which means that computation on these terms will generally get stuck and thus violate canonicity. Toward rectifying this, I prove a key identity regarding these axioms, add a further postulate asserting that this identity is equal to ref1, and convert both of these to rewrite rules:

```
rwEdgeConst1 : \forall {@b \ell : Level} {@b A : Set \ell} {a : A} \rightarrow (dA : isDiscrete A) \rightarrow edgeConst1 dA (eabs (\lambda \_ \rightarrow a)) \equiv (refl , refl) rwEdgeConst1 {a = a} dA = edgeConst2 dA (eabs (\lambda \_ \rightarrow a)) refl refl {-# REWRITE rwEdgeConst1 #-} postulate rwEdgeConst2 : \forall {@b \ell : Level} {@b A : Set \ell} {a : A} \rightarrow (dA : isDiscrete A) \rightarrow edgeConst2 dA (eabs (\lambda \_ \rightarrow a)) refl refl \equiv refl {-# REWRITE rwEdgeConst2 #-}
```

Although a full proof of canonicity is beyond the scope of this paper, I conjecture that adding these rules suffices to preserve canonicity, and I verify a few concrete cases of this conjecture later in the paper.

So much for the (weak) connectedness of I; let us now turn our attention to the other property we had previously stipulated of I, namely its *strict bipointedness*. As mentioned previously, we could simply postulate this stipulation directly as an axiom – however, for the purpose of proving parametricity theorems, a more prudent strategy is to instead formalize a class of I-indexed type families, whose computational behavior follows from this assumption (and which, in turn, implies it). Because these type families essentially correspond to the *graphs* of predicates and relations on arbitrary types, I refer to them as *graph types*.

Graph Types

I begin with an exposition of the simplest class of graph types: *unary graph types*, which, as the name would imply, correspond to graphs of unary predicates. Given a type A, a type family $B: A \to \mathsf{Type}$, and an element i: I, the *graph type* Gph^1 i A B is defined to be equal to A when i is i_0 , and equivalent to $\Sigma x: A.Bx$ when i is i_1 . Intuitively, an element of Gph^1 i A B is a dependent pair whose second element *only exists when* i *is equal to* i_1 . We may formalize this in Agda as follows, by postulating a rewrite rule that evaluates Gph^1 i_0 A B to A:

We then have the following introduction rule for elements of Gph^1 i A B, which are pairs where the second element of the pair only exists under the assumption that $i = i_1$. When $i = i_0$ instead, the pair

collapses to its first element:

```
g1pair : \forall \{\ell\} \{A : Set \ \ell\} \{B : A \to Set \ \ell\} (i : I)

\to (a : A) \to (b : (i \equiv i1) \to B \ a) \to Gph1 \ i \ A \ B

g1pair0 : \forall \{\ell\} \{A : Set \ \ell\} \{B : A \to Set \ \ell\}

\to (a : A) \to (b : (i0 \equiv i1) \to B \ a)

\to g1pair \{B = B\} \ i0 \ a \ b \equiv a

{-# REWRITE g1pair0 #-}
```

The first projection from such a pair may then be taken no matter what i is, and reduces to the identity function when i is i_0 :

The second projection, meanwhile, may only be taken when i is equal to i_1 :

```
g1snd : \forall {\ell} {A : Set \ell} {B : A \rightarrow Set \ell} 
 \rightarrow (g : Gph1 i1 A B) \rightarrow B (g1fst i1 g) 
g1beta2 : \forall {\ell} {A : Set \ell} {B : A \rightarrow Set \ell} 
 \rightarrow (a : A) \rightarrow (b : (i1 \equiv i1) \rightarrow B a) 
 \rightarrow g1snd (g1pair {B = B} i1 a b) \equiv b ref1 
{-# REWRITE g1beta2 #-}
```

It is straightforward to see that the inclusion of graph types makes strict bipointedness of the interval provable, as follows:

```
strBpt : (i1 \equiv i0) \rightarrow \bot
strBpt p = g1snd (transp (\lambda i \rightarrow Gph1 i \top (\lambda \rightarrow \bot)) p tt)
```

And in fact, the converse holds under the assumption of univalence. Specifically, in the presence of univalence and the assumption of strict bipointedness for I, the type Gph^1 i A B may be regarded as a computationally convenient shorthand for the type $\Sigma x : A.(i = i_1) \to Bx$, in much the same way as the type $\mathsf{Edge}\ A\ a_0\ a_1$ serves as shorthand for the type $\Sigma f : (\Pi i : I.Ai).(f\ i_0 = a_0) \times (f\ i_1 = a_1)$. This

fact is due to the following equivalence

```
\Sigma x : A.(i_0 = i_1) \to Bx
\simeq \Sigma x : A. \bot \to Bx
\simeq \Sigma x : A. \top
\simeq A
```

which, under univalence, becomes an identity between Σx : A.($i_0 = i_1$) $\rightarrow Bx$ and A, thereby justifying the use of this and associated identities as rewrite rules which, conjecturally, are fully compatible with canonictiy.

In addition to unary graph types, we also have *binary* graphs types for representing graphs of binary relations. That is, given types A, B, a type family $C: A \rightarrow B \rightarrow \mathsf{Type}$, and elements i, j: I, there is a type Gph^2 i j A B C. Intuitively, Gph^2 i j A B C is a type of dependent triples whose first element (of type A) exists only under the assumption that $i = i_0$, whose second element exists only under the assumption that $j = i_1$, and whose third element, which may depend upon the first and second, exists only under the conjunction of these assumptions. Hence Gph^2 i_0 i_0

postulate

```
Gph2 : \forall \{\ell\} (i j : I) (A : Set \ell) (B : Set \ell)
           \rightarrow (C : A \rightarrow B \rightarrow Set \ell) \rightarrow Set \ell
g2rw00 : \forall \{\ell\} (A : Set \ell) (B : Set \ell) (C : A \rightarrow B \rightarrow Set \ell)
              \rightarrow Gph2 i0 i0 A B C \equiv A
{-# REWRITE g2rw00 #-}
g2rw11 : \forall {\ell} (A : Set \ell) (B : Set \ell) (C : A \rightarrow B \rightarrow Set \ell)
              \rightarrow Gph2 i1 i1 A B C \equiv B
{-# REWRITE g2rw11 #-}
g2triple : \forall \{\ell\} (i j : I) {A : Set \ell} {B : Set \ell} {C : A \rightarrow B \rightarrow Set \ell}
                 \rightarrow (a : (i \equiv i0) \rightarrow A) \rightarrow (b : (j \equiv i1) \rightarrow B)
                 \rightarrow (c : (p : i \equiv i0) (q : j \equiv i1) \rightarrow C (a p) (b q))
                 → Gph2 i j A B C
g2triple00 : \forall {\ell} {A : Set \ell} {B : Set \ell} {C : A \rightarrow B \rightarrow Set \ell}
                     \rightarrow (a : (i0 \equiv i0) \rightarrow A) \rightarrow (b : (i0 \equiv i1) \rightarrow B)
                     \rightarrow (c : (p : i0 \equiv i0) (q : i0 \equiv i1) \rightarrow C (a p) (b q))
                     \rightarrow g2triple i0 i0 {C = C} a b c \equiv a refl
{-# REWRITE g2triple00 #-}
g2triple11 : \forall \{\ell\} \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
                     \rightarrow (a : (i1 \equiv i0) \rightarrow A) \rightarrow (b : (i1 \equiv i1) \rightarrow B)
                     \rightarrow (c : (p : i1 \equiv i0) (q : i1 \equiv i1) \rightarrow C (a p) (b q))
                     \rightarrow g2triple i1 i1 {C = C} a b c \equiv b refl
{-# REWRITE g2triple11 #-}
g2fst : \forall \{\ell\} (j : I) \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
            \rightarrow Gph2 iO j A B C \rightarrow A
```

```
g2beta1 : \forall \{\ell\} (j : I) {A : Set \ell} {B : Set \ell} {C : A \rightarrow B \rightarrow Set \ell}
               \rightarrow (a : (i0 \equiv i0) \rightarrow A) \rightarrow (b : (j \equiv i1) \rightarrow B)
               \rightarrow (c : (p : i0 \equiv i0) (q : j \equiv i1) \rightarrow C (a p) (b q))
               \rightarrow g2fst j (g2triple i0 j {C = C} a b c) \equiv a refl
{-# REWRITE g2beta1 #-}
g2fst00 : \forall \{\ell\} \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
               \rightarrow (g : Gph2 i0 i0 A B C) \rightarrow g2fst i0 {B = B} {C = C} g \equiv g
{-# REWRITE g2fst00 #-}
g2snd : \forall \{\ell\} (i : I) {A : Set \ell} {B : Set \ell} {C : A \rightarrow B \rightarrow Set \ell}
            \rightarrow Gph2 i i1 A B C \rightarrow B
g2beta2 : \forall \{\ell\} (i : I) {A : Set \ell} {B : Set \ell} {C : A \rightarrow B \rightarrow Set \ell}
               \rightarrow (a : (i \equiv i0) \rightarrow A) \rightarrow (b : (i1 \equiv i1) \rightarrow B)
               \rightarrow (c : (p : i \equiv i0) (q : i1 \equiv i1) \rightarrow C (a p) (b q))
               \rightarrow g2snd i (g2triple i i1 {C = C} a b c) \equiv b refl
{-# REWRITE g2beta2 #-}
g2snd11 : \forall \{\ell\} \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
               \rightarrow (g : Gph2 i1 i1 A B C) \rightarrow g2snd i1 {A = A} {C = C} g \equiv g
{-# REWRITE g2snd11 #-}
g2thrd : \forall \{\ell\} \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
              \rightarrow (g : Gph2 i0 i1 A B C) \rightarrow C (g2fst i1 g) (g2snd i0 g)
g2beta3 : \forall \{\ell\} \{A : Set \ell\} \{B : Set \ell\} \{C : A \rightarrow B \rightarrow Set \ell\}
               \rightarrow (a : (i0 \equiv i0) \rightarrow A) \rightarrow (b : (i1 \equiv i1) \rightarrow B)
               \rightarrow (c : (p : i0 \equiv i0) (q : i1 \equiv i1) \rightarrow C (a p) (b q))
               \rightarrow g2thrd (g2triple i0 i1 {C = C} a b c) \equiv c refl refl
{-# REWRITE g2beta3 #-}
```

Parametricity via Sufficient Cohesion

Some Applications

Impredicative Encodings

Modularity & Coinduction

Toward a synthetic theory of parametricity